

Object Oriented Programming and Perl

Prog for Biol 201 I
Simon Prochnik

Sunday, October 21, 12

1

Why do we teach you about objects and object-oriented programming (OOP)?

- Objects and OOP allow you to use other people's code to do a lot in just a few lines.
- For example, in the lecture on bioperl, you will see how to search GenBank by a sequence Accession, parse the results and reformat the sequence into any format you need in less than a dozen lines of object-oriented perl. Imagine how long it would take to write that code yourself!
- Someone else has already written and tested the code, so you don't have to.
- Most people don't ever write an object of their own: only create your own modules and objects if you have to
- search CPAN (www.cpan.org) to see if there is already a module that does what you need. There were 18,534 modules on Oct 14th 2010, this has grown to 100,575 (Oct 20, 2011), 114,367 Oct 19, 2012! Surely you can find a module to do what you want.

Sunday, October 21, 12

2

Using objects in perl


- some examples to show how you can use objects

Sunday, October 21, 12

3

Object-oriented programming is a programming style

- An object is a special kind of data structure (variable) that stores specific kinds of data and automatically comes with functions (methods) that can do useful things with that data
- Objects are often designed to work with data and functions that you would find associated with a real-world object or thing, for example, we might design gene sequence objects.
- A gene sequence object might store its chromosomal position and sequence data and have functions like transcribe() and new() to create a new object.



Sunday, October 21, 12

4

An example of a Microarray object that is designed specifically to handle microarray data

Tell perl you want to use objects in the Microarray class

Create a new object and load data

call the gene() subroutine to get gene name data from the object

call the tissue() subroutine to get tissue data from the object

```
#!/usr/bin/perl
#File: 00_script.pl
use strict;
use warnings;
use Microarray; # I wrote this example object class
my $microarray = Microarray->new( gene => 'CDC2',
                                  expression => 45,
                                  tissue => 'liver',
                                  );
my $gene_name = $microarray->gene();
print "Gene for this microarray is $gene_name\n";
my $tissue = $microarray->tissue();
print "The tissue is $tissue\n";
```

Output on screen:
Gene for this microarray is CDC2
The tissue is liver

Sunday, October 21, 12

5

An example that deals with statistics (Statistics::Descriptive objects)

Make new object with new()

Add data

Calculate mean

Calculate variance

```
#!/usr/bin/perl
#File: mean_and_variance.pl
use strict;
use warnings;
use Statistics::Descriptive; # this is on cpan.org
# need to make new object with S::D::Full->new()
my $stat = Statistics::Descriptive::Full->new();
$stat->add_data(1,2,3,4);
my $mean = $stat->mean();
my $var = $stat->variance();
print "mean is $mean\n";
print "variance is $variance\n";
```

Output on screen:
mean is 2.5
variance is 1.66666666666667

Sunday, October 21, 12

6

An example that deals with statistics (Statistics::Descriptive objects)

Make new object
with new()

Add data

Calculate mean

Calculate variance

```
#!/usr/bin/perl
#File: mean_and_variance.pl
use strict;
use warnings;

use Statistics::Descriptive;

my $stat = Statistics::Descriptive->new();
$stat->add_data(1,2,3,4);
my $mean = $stat->mean();
my $var = $stat->variance();
print "mean is $mean\n";
print "variance is $variance\n";
```

Output on screen:

mean is 2.5

variance is 1.66666666666667

Sunday, October 21, 12

7

Let's look at the new OOP syntax in more detail

```
# tell perl you want to use objects
# in a certain class
use Statistics::Descriptive;
```


Here's the class name
'Statistics::Descriptive'. perl will look for a
module with the filename
.../Statistics/Descriptive.pm

Sunday, October 21, 12

8

Let's look at the new OOP syntax in more detail

Before you can use an object, you create one.
This is often done with a call to a new() method.




Sunday, October 21, 12

9

Let's look at the new OOP syntax in more detail

Once you have created an object you call methods on it
to use the object



Sunday, October 21, 12

10

Object-oriented programming in a little more detail

- Let's look at which elements of perl are used to provides object oriented programming

Sunday, October 21, 12

11

Object Oriented Programming and Perl

- To understand object-oriented syntax in perl, we need to recap three things: **references, subroutines, packages**.
- These three elements of perl are recycled with slightly different uses to provide object-oriented programming

What you can do	Normal perl (procedural perl)	Object-oriented perl
organize code that goes together for reuse	package	class (the type or kind of object, and all the code that goes with it)
store data (simple or very complex)	a reference	the object itself (a reference to a data structure)
work on data by writing simple code	subroutine	a method (function that acts on the object)

Sunday, October 21, 12

12

Object Oriented Programming and Perl

- The OOP paradigm provides i) a solid framework for sharing code -- reuse
- and ii) a guarantee or contract or specification for how the code will work and how it can be used -- an interface
- and iii) hides the details of implementation so you only have to know how to use the code, not how it works -- saves you time, quick to learn, harder to introduce bugs
- Here we are briefly introducing you to OOP and objects so that you can quickly add code that's already written into your scripts, rather than spend hours re-inventing wheels. Many more people use objects than write them.

Sunday, October 21, 12

13

I: Recap references

example of syntax

```
$ref_to_hash = {key1=>'value1',key2=>'value2',...}
```

code example


```
my $microarray = {gene => 'CDC2',  
                  expression => 45,  
                  tissue => 'liver',  
                  };
```

We can store any pieces of data we would like to keep together in a hash

Here is the data structure in memory

scalar hash reference

`$microarray`



anonymous hash

key	value
gene	CDC2
expression	45
tissue	liver

Sunday, October 21, 12

14

II: recap subroutines

- solve a problem, write code once, and call the code simply
- reusing a single piece of code instead of copying, pasting and modifying reduces the chance you'll make an error and simplifies bug fixing.

```
#!/usr/bin/perl -w
use strict;
my $seq;
while (my $seqline = <>) { # read sequence from standard in
    my $clean = cleanup_sequence($seqline); # clean it up
    $seq      .= $clean;                  # add it to full sequence
}
sub cleanup_sequence {
    my ($sequence) = @_; # set $sequence to first argument
    $sequence = lc $sequence; # translate everything into lower case
    $sequence =~ s/[\s\d]//g; # remove whitespace and numbers
    $sequence =~ m/^[gactcn]+$/ or die "Sequence contains invalid
                                characters!";
    return $sequence;
}
```

Sunday, October 21, 12

15

III: now let's recap packages

- organise code that goes together into reusable modules, packages

```
#!/usr/bin/perl -w                                     read_clean_sequence.pl
#File: read_clean_sequence.pl
use strict;
use Sequence;
my $seq;
while (my $seqline = <>) { # read sequence from standard in
    my $clean = cleanup_sequence($seqline); # clean it up
    $seq      .= $clean;                  # add it to full sequence
}
```

```
#file: Sequence.pm                                     Sequence.pm
package Sequence;
use strict;
use base Exporter;
our @EXPORT = ('cleanup_sequence');
sub cleanup_sequence {
    my ($sequence) = @_; # set $sequence to first argument
    $sequence = lc $sequence; # translate everything into lower case
    $sequence =~ s/[\s\d]//g; # remove whitespace and numbers
    $sequence =~ m/^[gactcn]+$/ or die "Sequence contains invalid
characters!";
    return $sequence;
}
1;
```

Sunday, October 21, 12

16

Let's recap subroutines: new example with references

```
#!/usr/bin/perl
use strict;
use warnings;
my $microarray = { gene => 'CDC2',
                  expression => 45,
                  tissue => 'liver',
                  };
...
my $gene_name = gene($microarray);
...
sub gene {
    my ($ref) = @_;
    return $$ref{gene};
}
sub tissue {
    my ($ref) = @_;
    return $$ref{tissue};
}
```

Sunday, October 21, 12

17

recap packages

main script
file

```
#!/usr/bin/perl                                script.pl
#File: script.pl
use strict; use warnings;
use Microarray;

my $microarray = {gene => 'CDC2',
                  expression => 45,
                  tissue => 'liver',
                  };
my $gene_name = gene($microarray);
print "Gene for this microarray is
$gene\n";
```

perl module file

```
Microarray.pm
#File: Microarray.pm
package Microarray;
use strict;
use base Exporter;

our @EXPORT = ('gene', 'tissue');

sub gene {
    my ($ref) = @_;
    return $$ref{gene};
}
sub tissue {
    my ($ref) = @_;
    return $$ref{tissue};
}
1;
```

Sunday, October 21, 12

18

Let's look at how you create object code


- This is mostly for reference.
- You'll probably use it rarely, if at all

Sunday, October 21, 12

19

Three Little Rules

- Rule 1: To create a class, build a package
- Rule 2: To create a method, write a subroutine
- Rule 3: To create an object, bless a reference



Sunday, October 21, 12

20

Rule 1: To create a class, build a package

- all the code that goes with an object (methods, special variables) goes inside a special package
 - perl packages are just files whose names end with '.pm' e.g. `Microarray.pm`
 - package filenames should start with a capital letter
 - the name of the perl package tells us the class of the object. This is really the type or kind of object we are dealing with.
- `Micorarray.pm` is a package, so it will be easy to convert into object-oriented code

Sunday, October 21, 12

21

Rule 2: To create a method, write a subroutine

- we already have `gene()` in `Microarray.pm`
- this can be turned into a method
- we need one extra subroutine to create new objects
- the creator method is called `new()` and has one piece of magic...

Sunday, October 21, 12

22

Rule 3: To create an object, bless a reference

- The `new()` subroutine uses the `bless` function to create an object
- full details coming up... but here's the skeleton of a `new()` method

```
sub new {  
    ...  
    my $self = {};  
    bless $self, $class;  
    ...  
}
```

create a reference, a
hashref {} is the most
common seen in perl

bless a reference
into a class

Sunday, October 21, 12

23

Let's recap packages

```
#!/usr/bin/perl -w  
#File: script.pl  
use strict;  
use Microarray;  
  
my $microarray = { gene => 'CDC2',  
                  expression => 45,  
                  tissue => 'liver',  
                  };  
my $gene_name = gene($microarray);  
print "Gene for this microarray is $gene\n";
```

```
#File: Microarray.pm  
package Microarray;  
use strict;  
use base Exporter;  
  
our @EXPORT = ('gene', 'tissue');  
  
sub gene {  
    my $ref = shift;  
    return $$ref{gene};  
}  
sub tissue {  
    my $ref = shift;  
    return $$ref{tissue};  
}  
1;
```

Sunday, October 21, 12

24

Transforming a package into an object-oriented module or class

procedural perl package
(what you saw yesterday)

...transforming the package into a class...

```
#File: Microarray.pm
package Microarray;
use strict;
use base Exporter;

our @EXPORT = ('gene', 'tissue');

sub gene {
    my ($ref) = @_;
    return ${$ref}{gene};
}
sub tissue {
    my ($ref) = @_;
    return ${$ref}{tissue};
}
1;
```



```
#File: Microarray.pm
package Microarray;
use strict;

sub gene {
    my $self = shift; # same as my ($self) = @_;
    return ${$self}{gene};
}
sub tissue {
    my $self = shift;
    return ${$self}{tissue};
}
1;
```

Sunday, October 21, 12

25

The new() method is a subroutine that creates a new object

```
sub new {
    my $class = shift;
    my %args = @_;
    my $self = {};
    foreach my $key (keys %args) {
        ${$self}{$key} =
            $args{$key};
    }
    # the magic happens here
    bless $self, $class;
    return $self;
}
```

the first argument is always the class of the object you are making. **perl gives you this as the first argument automatically**

a hash reference is the data structure you build an object from in perl

here we initialize variables in the object (in case there are any)

Some people like to write

```
`${$self}{$key}
```

as

```
$self -> {$key}
```

bless makes the object \$self (which is a hash reference) become a member of the class \$class

Sunday, October 21, 12

26

bless creates an object by making a reference belong to a class

Make an anonymous hash in the debugger

```
$a = {};  
p ref $a;  
HASH
```

Make a MySequence object in the debugger

```
$self = {};  
$class = 'MySequence';  
bless $self , $class;  
  
x $self  
0 MySequence=HASH(0x18bd7cc)  
  empty hash  
p ref $a  
MySequence
```

Sunday, October 21, 12

27

final step

object-oriented module or class

```
#File: Microarray.pm  
package Microarray;  
use strict;  
  
sub new {  
  my $class = shift;  
  my %args = @_;  
  my $self = {};  
  foreach my $key (keys %args) {  
    ${$self}{$key} = $args{$key};  
  }  
  # the magic happens here  
  bless $self, $class;  
  return $self;  
}  
  
sub gene {  
  my $self = shift;  
  return ${$self}{gene};  
}  
  
sub tissue {  
  my $self = shift;  
  return ${$self}{tissue};  
}  
1;
```

Sunday, October 21, 12

28

OOP script

```
#!/usr/bin/perl
use strict; use warnings;          procedural version
#File: script.pl
my $microarray = { gene => 'CDC2',
                  expression => 45,
                  tissue => 'liver',
                  };
my $gene_name = gene($microarray);
print "Gene for this microarray is $gene\n";
```

```
#!/usr/bin/perl
#File: OO_script.pl
use strict; use warnings;
use Microarray;
my $microarray = Microarray->new( gene => 'CDC2',
                                 expression => 45,
                                 tissue => 'liver',
                                 );
my $gene_name = $microarray->gene();
print "Gene for this microarray is $gene_name\n";
my $tissue = $microarray->tissue();
print "The tissue is $tissue\n";
```

Sunday, October 21, 12

29

Lastly, did I mention “code lazy”?

- This lecture has introduced you to object-oriented programming
- You only need to **use** other people’s objects (beg, borrow, buy, steal).
- Only create your own modules and objects if you **have to**.

Sunday, October 21, 12

30

Problems

1. Take a look at the Statistics::Descriptive module on cpan here <http://search.cpan.org/~shlomif/Statistics-Descriptive-3.0202/lib/Statistics/Descriptive.pm>

2. Write a script that uses the methods in Statistics::Descriptive to calculate the standard deviation, median, min and max of the following numbers

12,-13,-12,7,11,-4,-12,9,6,7,-9

Optional questions

4. Add a method to Microarray.pm called expression() which returns the expression value

5. Currently calling \$a = \$m->gene() gets the value of gene in the object \$m. Modify the gene() method so that if you call gene() with an argument, it will set the value of gene to be that argument e.g.

```
$m->gene('FOXP1');           # this should set the
                              #gene name to 'FOXP1'
print $m->gene();             # this should print the value 'FOXP1'
```

Further reading on inheritance

- If you want to make an object that is a special case or subclass of another, more general, object, you can have it inherit all the general data storage and functions of the more general object.
- This saves coding time by re-using existing code. This also avoids copying and pasting existing code into the new object, a process that makes code harder to maintain and debug.
- For example, a MicroRNA_gene object is a special case of a Gene object and might have some specific functions like cut_RNA_hairpin() as well as general functions like transcribe() it can **inherit** from the general gene object.
- More formally, a subclass inherits variables and functions from its superclass (like a child and a parent). Here are some examples

```
package MicroRNA;
use base 'Gene'; # Gene is a parent
use base 'Exporter'; # Exporter is another parent
```